

# Tema 8:

# Optimización de procesos industriales

- Objetivos del tema.
- Problemas de Planificación de la Producción.
- Planificación de la producción de electricidad en centrales térmicas.
- Planificación de la producción de diferentes derivados en una refinería.

### Objetivos del tema:


- Dar una introducción al tipo de problemas que se pretenden resolver
- Presentar un par de ejemplos de problemas de optimización de un proceso industrial
- Proponer un problema similar como proyecto en el que el alumno debe:
  - Desarrollar un modelo matemático del mismo
  - Implementarlo mediante la herramienta ILOG OPL Studio
  - Obtener la solución óptima en un escenario concreto
  - Opcionalmente, se pueden proponer mejoras en el modelo para contemplar un mayor número de características

## Problemas de Planificación de la Producción (1)

La **planificación de la producción** en plantas de proceso es uno de los problemas más complejos e importantes para una amplia variedad de **procesos industriales**.

Para el caso de plantas de proceso discontinuo el problema en planificar la producción sobre un horizonte de planificación discreto. Este tipo de problemas se conocen en la literatura como problemas DLSP (Discrete Lot Sizing and Scheduling Problems).

En el caso más sencillo, un productor fabrica un artículo cuya demanda varía en el tiempo, de acuerdo con el gráfico de la figura, donde el tiempo pueden ser horas, días, semanas e incluso meses.



Para asegurar que la demanda es convenientemente atendida en cada periodo, el productor tiene en general dos alternativas:

### 1. Producción variable:

El fabricante puede producir en cada periodo el número exacto de unidades que le solicitan.

Sin embargo, producción que varía en el tiempo es costosa de mantener debido a los costes asociados a los horarios más largos (horas extra) en los periodos de producción alta y los costes asociados al paro del personal y la maquinaria en los periodos de producción baja. Además, podría ser más eficiente producir stock extra en los periodos con coste de producción más bajo.

### 2. Producción constante:

Dado que la demanda en cada periodo puede satisfacerse bien a partir de la producción en dicho periodo o bien a a partir del stock almacenado, el fabricante puede producir por encima de la demanda en periodos de baja demanda y almacenar la sobreproducción para satisfacer la demanda los periodos de demanda mayor. Así, la producción puede mantenerse constante.

Sin embargo, tal opción puede no ser deseable si los costes de almacenamiento del stock extra son demasiado altos.

## Problemas de Planificación de la Producción (2)

Como puede verse, no es fácil decidir cual de las dos opciones anteriores es mejor ya que cualquiera de ellas puede dar lugar a una utilización ineficiente del capital. La primera opción podría dar lugar a un aumento de los costes de producción pero minimizaría los costes de inventario ya que sólo se produce la demanda, mientras que la segunda podría conducir a un exceso de inventario pero no incrementaría los costes de producción debidos a horas extra y paros de maquinaria.

Los problemas de esta naturaleza ilustran las dificultades que surgen cuando objetivos contrarios están presentes en un sistema dado. Así, en su versión más sencilla, el objetivo es llevar a cabo una planificación de la producción que satisfaga la demanda en cada periodo minimizando los costes asociados a las variaciones en la producción y al almacenamiento de exceso de stock en los almacenes.

**Ejemplo:** Supóngase un stock inicial de  $A_0=2$  artículos y la siguiente demanda a lo largo de  $T=4$  periodos  $D=[2, 3, 6, 1]$ . Además, se conoce el coste de producción en cada periodo  $CP=[1, 1, 2, 3]$  y el coste de almacenamiento  $CA=[2, 2, 2, 2]$ . Tanto la capacidad de producción como la de almacenamiento están limitadas a  $P_{MAX}=5$  y  $A_{MAX}=10$  respectivamente. Se trata de encontrar el número de artículos a producir en cada periodo, sin exceder la capacidad de producción y de almacenamiento de stock en cada periodo, de tal forma que se satisfaga la demanda y se minimicen los costes de producción y de inventario.

Para llevar a cabo la formulación matemática del modelo es necesario definir los conjuntos y parámetros del modelo. A partir del enunciado, vemos que éstos son los siguientes:

### Conjuntos

$T$           Número de periodos en el horizonte de planificación indexado en  $t=1, \dots, T$

### Parámetros

$A_0$           Nivel de inventario al inicio del horizonte de planificación

$A_{MAX}$         Máxima capacidad de almacenamiento

$P_{MAX}$         Máxima capacidad de producción

$D_t$           Demanda en el periodo  $t$

$CP_t$          Coste de producción en el periodo  $t$

$CA_t$          Coste de almacenamiento en el periodo  $t$

### Problemas de Planificación de la Producción (3)

Ahora, nótese que necesitamos una variable para calcular la solución al problema (número de unidades a producir en cada periodo) y otra para controlar el nivel de inventario en los almacenes (ya que de dicho nivel depende el coste de inventario que hay que minimizar, y sobre dicha variable se deberán imponer las restricciones sobre la capacidad de almacenamiento). Así, las variables del problema son:

#### Variables

$p_t$  Producción en el periodo  $t$

$a_{t,p}$  Nivel de inventario (número de unidades almacenadas) en el periodo  $t$

Una vez definidos los conjuntos, parámetros y variables del problema, ya se puede proceder a la especificación del modelo matemático que nos dará la solución óptima del problema. Dicho modelo estará compuesto por una función objetivo más un conjunto de restricciones.

#### Función objetivo:

Se trata de minimizar los costes de producción y almacenamiento.

$$(1) \quad \min z = \sum_{t=1}^T CP_t \cdot p_t + \sum_{t=1}^T CA_t \cdot a_t$$

#### Restricciones:

Tanto la capacidad de producción como la capacidad de almacenamiento están limitadas:

$$(2a) \quad 0 \leq p_t \leq PMAX \quad ; \quad \forall(t)$$

$$(2b) \quad 0 \leq a_t \leq AMAX \quad ; \quad \forall(t)$$

**Balance de material:** el nivel de inventario en cada periodo es igual al nivel en el periodo anterior más la producción en el periodo actual menos la demanda en dicho periodo. La segunda ecuación es la primera pero particularizada para el primer periodo.

$$(3a) \quad a_t = a_{t-1} + p_t - D_t \quad ; \quad \forall(t) : t > 1$$

$$(3b) \quad a_1 = A0 + p_1 - D_1 \quad ;$$

## Problemas de Planificación de la Producción (4)

Implementación en ILOG OPL Studio (modelo + datos):

```
// Conjuntos modelo.mod
int NT = ...;
setof(int) T = {t | t in 1..NT};

// Parámetros
int A0 = ...;
int AMAX = ...;
int PMAX = ...;
int D[T] = ...;
int CP[T] = ...;
int CA[T] = ...;

// Variables
dvar int+ p[T];
dvar int+ a[T];
dvar float+ obj1;
dvar float+ obj2;

// Etiquetado de restricciones
constraint C2a;
constraint C2b;
constraint C3a;
constraint C3b;
```

```
// Dimensiones de los conjuntos
NT = 4;
```

```
// Valores de los parámetros
```

```
A0 = 2;
AMAX = 5;
PMAX = 10;
D = [ 2, 3, 6, 1 ];
CP = [ 1, 1, 4, 3 ];
CA = [ 2, 2, 2, 2 ];
```

**datos.dat**

```
// Función objetivo modelo.mod (continuación)
minimize obj1 + obj2;


// Restricciones
subject to {

// 1. Función objetivo
obj1==sum(t in T) CP[t]*p[t];
obj2==sum(t in T) CA[t]*a[t];

// 2. Capacidades máximas
C2a = forall(t in T) p[t]<=PMAX;
c2b = forall(t in T) a[t]<=AMAX;

// 3. Balance de material
C3a = forall(t in T: t>1) a[t]==a[t-1]+p[t]-D[t];
C3b = a[1]==A0 +p[1]-D[1];

}
```



```
Final solution with objective 25:
p = [0 8 1 1];
a = [0 5 0 0];
```

**Solución**

## Planificación de la producción de electricidad en centrales térmicas (1)

Antes de pasar al problema propuesto como ejercicio de este tema, veamos otro problema de planificación de la producción de mayor complejidad que el anterior.

En el siguiente nivel de dificultad, pueden considerarse  $K$  productores que pueden fabricar un producto simultáneamente para satisfacer la demanda. Además, existen otras características adicionales como:

- Un coste asociado a la parada ( $CE_k$ ) o puesta en marcha ( $CS_k$ ) de la maquinaria de cada productor, más un coste de producción fijo asociado al funcionamiento de dicha maquinaria ( $CPF_k$ ), además del coste variable de producción ( $CVP_k$ ).
- Además de un límite en la máxima capacidad de producción por periodo de cada productor ( $PMAX_k$ ), puede existir un límite inferior ( $PMIN_k$ ), ya que podría no merecer la pena poner en funcionamiento una máquina si no es para producir una cantidad suficiente de producto.
- Adicionalmente, la máxima variación en la producción entre dos periodos consecutivos de cada productor está limitada, tanto para un incremento ( $PIMAX_k$ ) como para un decremento ( $PDMIN_k$ ) en la producción.
- También podemos encontrar límites inferiores en la capacidad de almacenamiento ( $AMIN_k$ ) que aseguren un stock mínimo para satisfacer algún imprevisto en la demanda.
- Respecto a este último punto, también se puede imponer que la suma de la máxima capacidad de producción de todos los productores sea superior a la demanda ( $D_k$ ) de cada periodo en una determinada cantidad denominada reserva ( $R_k$ ).

Este tipo de problemas encaja perfectamente en el problema de la **programación de centrales térmicas de producción de electricidad**.

El coste de poner en funcionamiento una central térmica de producción de energía eléctrica, habiendo estado parada un par de días, es del orden del coste de compra de un apartamento en una buena zona residencial de una ciudad media. Por tanto, la planificación de los arranques y paradas de las centrales térmicas ha de hacerse con cuidado.

El problema de la programación horaria de centrales térmicas consiste en determinar para un horizonte de planificación multi-horario, el arranque y parada de cada central, de tal forma que se suministre la demanda en cada hora, el coste se minimice, y se satisfagan determinadas restricciones técnicas y de seguridad.

## Planificación de la producción de electricidad en centrales térmicas (2)

Supóngase que se desea planificar la producción de electricidad en  $K$  centrales térmicas a lo largo de un horizonte de planificación de  $T$  horas. Los datos del problema son las demandas en esas horas, así como la cantidad de reserva en las mismas. Se conoce además el estado de todas las centrales (paradas o funcionando) en el periodo previo al comienzo del horizonte de planificación. Además se dispone de los costes de arranque, parada, producción fija y variable, así como de los límites en la capacidad de producción de cada central y en el incremento y decremento en la producción por periodo.

De nuevo, para llevar a cabo la formulación matemática del modelo es necesario definir los conjuntos y parámetros del modelo. A partir del enunciado, vemos que éstos son los siguientes:

### Conjuntos

$T$  Número de periodos en el horizonte de planificación indexado en  $t=1,\dots,T$

$K$  Número de centrales disponibles indexado en  $k=1,\dots,K$

### Parámetros

$D_t$  Demanda en el periodo  $t$

$R_t$  Reserva requerida en el periodo  $t$

$X0_k$  Denota si la central  $k$  está funcionando al inicio del horizonte de planificación

$P0_k$  Producción de la central  $k$  al inicio del horizonte de planificación

$CPF_k$  Coste de producción fijo de la central  $k$

$CPV_k$  Coste de producción variable de la central  $k$

$CS_k$  Coste de arranque de la central  $k$

$CE_k$  Coste de parada de la central  $k$

$PMIN_k$  Producción mínima de la central  $k$

$PMAX_k$  Producción máxima de la central  $k$

$PIMAX_k$  Máximo incremento por periodo en la producción de la central  $k$

$PDMAX_k$  Máximo decremento por periodo en la producción de la central  $k$

### Planificación de la producción de electricidad en centrales térmicas (3)

Ahora, nótese que necesitamos una variable para calcular la solución al problema (cantidad de electricidad a producir en cada periodo  $t$  por cada central  $k$ ). Dicha variable es además la que nos ayudará a escribir el término asociado al coste de producción variable ( $CPV_k$ ).

Por otro lado, existe un coste de producción fijo ( $CPF_k$ ) asociado al simple hecho de que una central esté funcionando o no, por lo que podemos definir una variable binaria ( $x_{t,k}$ ) para que tome el valor 1 si la central  $k$  está funcionando en el periodo  $t$  y 0 en caso contrario para escribir el término de la función objetivo asociado a dicho coste.

Análogamente, los términos asociados al coste por la puesta en marcha y las paradas de las centrales en cada periodo pueden escribirse con ayuda de un par de variables binarias que representen la ocurrencia de dichos hechos. Por ejemplo,  $s_{t,k}$  con valor 1 si la central  $k$  arranca al comienzo del periodo  $t$  y 0 en caso contrario, y  $e_{t,k}$  con valor 1 si la central  $k$  para al comienzo del periodo  $t$  y 0 en caso contrario. Así, las variables del problema son:

#### Variables

- $p_{t,k}$  Producción de la central  $k$  durante el periodo  $t$
- $x_{t,k}$  Denota si la central  $k$  se está funcionando durante el periodo  $t$
- $s_{t,k}$  Denota si la central  $k$  arranca al comienzo del periodo  $t$
- $e_{t,k}$  Denota si la central  $k$  para al comienzo en el periodo  $t$

Una vez definidos los conjuntos, parámetros y variables del problema, ya se puede proceder a la especificación del modelo matemático que nos dará la solución óptima del problema. Dicho modelo estará compuesto por una función objetivo más un conjunto de restricciones.

#### Función objetivo:

Se trata de minimizar los costes de producción (fijo y variable) más los asociados a la puesta en marcha y paradas de las centrales a lo largo de todo el horizonte de planificación.

Así, teniendo en cuenta la definición de dichos costes en el enunciado del problema y la definición de las variables anterior, dicha función objetivo puede escribirse como:

$$(1) \quad \min z = \sum_{t=1}^T \sum_{k=1}^K CPF_k \cdot x_{t,k} + \sum_{t=1}^T \sum_{k=1}^K CPV_k \cdot p_{t,k} + \sum_{t=1}^T \sum_{k=1}^K CS_k \cdot s_{t,k} + \sum_{t=1}^T \sum_{k=1}^K CE_k \cdot e_{t,k}$$

## Planificación de la producción de electricidad en centrales térmicas (4)

### Restricciones:

Las centrales térmicas no pueden funcionar ni por debajo de una producción mínima, ni por encima de una producción máxima. Estas restricciones se pueden formular tal y como se muestra en la ecuación (2). El término de la izquierda establece que si la central  $k$  está funcionando durante el periodo  $t$  ( $x_{t,k}=1$ ), su producción ( $p_{t,k}$ ) ha de estar por encima de su producción mínima. De forma análoga, el término de la derecha hace que si la central  $k$  está funcionando durante el periodo  $t$ , su producción ha de estar por debajo de su producción máxima. Nótese además que si  $x_{t,k}=0$ , se fuerza a que  $p_{t,k}=0$ .

$$(2) \quad PMIN_k \cdot x_{t,k} \leq p_{t,k} \leq PMAX_k \cdot x_{t,k} \quad ; \quad \forall(t,k)$$

Al pasar de un periodo de tiempo al siguiente, cualquier central térmica no puede incrementar su producción por encima de un máximo, denominado rampa máxima de subida de carga. Esta restricción se expresa mediante la ecuación (3a). La ecuación (3b) es la particularización de la (3a) para el primer periodo.

$$(3a) \quad p_{t,k} - p_{t-1,k} \leq PIMAX_k \quad ; \quad \forall(t,k) : t > 1$$

$$(3b) \quad p_{1,k} - P0_k \leq PIMAX_k \quad ; \quad \forall(k)$$

Análogamente, ninguna central puede bajar su producción por encima de un máximo, que se denomina rampa máxima de bajada de carga. Esta restricción se expresa mediante la ecuación (4a). La ecuación (4b) es la particularización de la (4a) para el primer periodo.

$$(4a) \quad p_{t-1,k} - p_{t,k} \leq PDMAX_k \quad ; \quad \forall(t,k) : t > 1$$

$$(4b) \quad P0_k - p_{1,k} \leq PDMAX_k \quad ; \quad \forall(k)$$

Cualquier central que está funcionando puede pararse pero no arrancarse, y análogamente cualquier central parada puede arrancarse pero no pararse. Así, la lógica de cambio de estado (de arranque a parada y viceversa) ha de preservarse a lo largo de todo el horizonte de planificación. Esta restricción se expresa mediante la ecuación (4a) para  $t > 1$  y mediante la ecuación (4b) para el primer periodo.

$$(5a) \quad s_{t,k} - e_{t,k} = x_{t,k} - x_{t-1,k} \quad ; \quad \forall(t,k) : t > 1$$

$$(5b) \quad s_{1,k} - e_{1,k} = x_{1,k} - X_k \quad ; \quad \forall(k)$$

## Planificación de la producción de electricidad en centrales térmicas (5)

La demanda debe suministrarse en cada periodo.

$$(6) \quad \sum_{k=1}^K P_{t,k} = D_k \quad ; \quad \forall(t)$$

Finalmente, por razones de seguridad, la potencia total disponible en centrales en funcionamiento debe ser mayor que la demanda en una determinada cantidad de reserva.

$$(7) \quad \sum_{k=1}^K PMAX_k \cdot x_{t,k} \geq D_k + R_k \quad ; \quad \forall(t)$$

**Ejemplo:** Supóngase que se desea planificar la producción de electricidad en  $K=3$  centrales térmicas a lo largo de un horizonte de planificación de  $T=5$  horas. Las demandas en esas horas es  $D=[150, 500, 400, 350, 200]$ . Además, la reserva en dichas horas es  $R=[15, 50, 40, 50, 30]$ . Se considera además que todas las centrales están paradas en el periodo previo al comienzo del horizonte de planificación,  $X_0=[0, 0, 0]$ . Además se dispone de los costes de arranque, parada, producción fija y variable, así como de los límites en la capacidad de producción de cada central y de los límites máximos de incremento y decremento en la producción por periodo de cada central. Estos datos se recogen en la siguiente tabla.

	Central		
Parámetro	No 1	No 2	No 3
Coste de producción fijo	5	7	6
Coste de producción variable	0.100	0.125	0.150
Coste de arranque	20	18	5
Coste de parada	0.5	0.3	0.1
Producción mínima	50	80	40
Producción máxima	350	200	140
Rampa máxima de subida de carga	200	100	100
Rampa máxima de bajada de carga	300	150	100

## Planificación de la producción de electricidad en centrales térmicas (6)

Implementación en ILOG OPL Studio: Modelo:

```
// Conjuntos                                modelo.mod
int NT = ...;
int NK = ...;
setof(int) T = {t | t in 1..NT};
setof(int) K = {k | k in 1..NK};

// Parámetros
float D[T] = ...;
float R[T] = ...;
int X0[K] = ...;
float P0[K] = ...;
float CPF[K] = ...;
float CPV[K] = ...;
float CS[K] = ...;
float CE[K] = ...;
float PMAX[K] = ...;
float PMIN[K] = ...;
float PIMAX[K] = ...;
float PDMAX[K] = ...;

// Variables
dvar boolean x[T,K];
dvar boolean s[T,K];
dvar boolean e[T,K];
dvar float+ p[T,K];
dvar float+ obj1;
dvar float+ obj2;
dvar float+ obj3;
dvar float+ obj4;
```

```
// Restricciones                                modelo.mod (continuación 1)
constraint C2a;
constraint C2b;
constraint C3a;
constraint C3b;
constraint C4a;
constraint C4b;
constraint C5a;
constraint C5b;
constraint C6;
constraint C7;
```

## Planificación de la producción de electricidad en centrales térmicas (7)

```
// Función objetivo
minimize obj1 + obj2 + obj3 + obj4;

// Restricciones
subject to {

// 1. Función objetivo
obj1==sum(t in T, k in K) CPF[k]*x[t,k];
obj2==sum(t in T, k in K) CPV[k]*p[t,k];
obj3==sum(t in T, k in K) CS[k]*s[t,k];
obj4==sum(t in T, k in K) CE[k]*e[t,k];

// 2. Límites en la producción
C2a = forall(t in T, k in K) p[t,k]>=PMIN[k]*x[t,k];
C2b = forall(t in T, k in K) p[t,k]<=PMAX[k]*x[t,k];

// 3. Máximo incremento en la producción por periodo
C3a = forall(t in T: t>1, k in K) p[t,k]-p[t-1,k]<=PIMAX[k];
C3b = forall(
           k in K) p[1,k]-P0[k] <=PIMAX[k];

// 4. Mínimo incremento en la producción por periodo
C4a = forall(t in T: t>1, k in K) p[t-1,k]-p[t,k]<=PDMAX[k];
C4b = forall(
           k in K) P0[k] -p[1,k]<=PDMAX[k];

// 5. Lógica del cambio de estado
C5a = forall(t in T: t>1, k in K) s[t,k]-e[t,k]==x[t,k]-x[t-1,k];
C5b = forall(
           k in K) s[1,k]-e[1,k]==x[1,k]-X0[k];

// 6. Demanda
C6 = forall(t in T) sum(k in K) p[t,k]==D[t];

// 7. Reserva
C7 = forall(t in T) sum(k in K) PMAX[k]*x[t,k]>=D[t]+R[t];

}
```

**modelo.mod (continuación 2)**

## Planificación de la producción de electricidad en centrales térmicas (8)

Implementación en ILOG OPL Studio: Datos

```
// Dimensiones de los conjuntos
NT = 5;
NK = 3;

// Valores de los parámetros
D = [ 150 , 500 , 400 , 350 , 200 ];
R = [ 15 , 50 , 40 , 50 , 30 ];
X0 = [ 0 , 0 , 0 ];
P0 = [ 0 , 0 , 0 ];
CPF = [ 5 , 7 , 6 ];
CPV = [ 0.100, 0.125, 0.150 ];
CS = [ 20 , 18 , 5 ];
CE = [ 0.5 , 0.3 , 1.0 ];
PMIN = [ 50 , 80 , 40 ];
PMAX = [ 350 , 200 , 140 ];
PIMAX = [ 200 , 100 , 100 ];
PDMAX = [ 300 , 150 , 100 ];
```

**datos.dat**

### Solución

Final solution with objective 263.8:

```
obj1 = 50;
obj2 = 169.5;
obj3 = 43;
obj4 = 1.3;
```


```
x = [[1 0 0]
      [1 1 1]
      [1 0 1]
      [1 0 1]
      [1 0 0]];
```

```
p = [[150 0 0]
      [350 100 50]
      [350 0 50]
      [310 0 40]
      [200 0 0]];
```

```
s = [[1 0 0]
      [0 1 1]
      [0 0 0]
      [0 0 0]
      [0 0 0]];
```

```
e = [[0 0 0]
      [0 0 0]
      [0 1 0]
      [0 0 0]
      [0 0 1]];
```

Ejecución ILOG OPL Studio:



## Planificación de la producción de electricidad en centrales térmicas (9)

Finalmente, veamos otras alternativas para el **modelado de arranques y paradas**.

Otra posibilidad para modelar el cambio de estado en las centrales (especialmente útil cuando se producen varios productos en cada una de las plantas, como se verá en el proyecto propuesto) consiste en eliminar las variables  $s_k$  y  $e_k$  y modificar la definición de la variable  $x_k$ .

Para poder modelar los cambios de estado tan sólo a partir de la variable  $x$ , podemos añadir un índice adicional  $f$  indexado en cada uno de los diferentes estados en los que se pueda encontrar cada planta  $k$ . En el caso de las centrales, los posibles estados tan sólo son dos, parada ( $f=0$ ) ó funcionando ( $f=1$ ). Nótese, que de esta forma se puede generalizar el número de estados a más de dos (en general  $F$ ), parada ( $f=0$ ), produciendo el producto tipo 1 ( $f=1$ ), produciendo el producto tipo 2 ( $f=2$ ), etc ... Así, para cada central la variable  $x$  es un vector con un 1 en el índice asociado al estado en el que se encuentre dicha central. Evidentemente, cada central tan sólo se puede encontrar en un estado en cada periodo. Este hecho puede reflejarse en el modelo a través de la siguiente restricción.

$$(8') \quad \sum_{f=0}^{F-1} x_{t,k,f} = 1 \quad ; \quad \forall (t,k)$$

Por otro lado, para calcular el coste asociado al cambio de estado de cada central en periodos diferentes, se puede definir una variable de cambio de estado o reconfiguración de la central,  $cr_{t,k,t,f}$ , indexada en el periodo  $t$ , la central  $k$ , y los estados  $f$  y  $f'$  en los que se pueda encontrar dicha central en los periodos  $t$  y  $t-1$  respectivamente.

Así, la nueva definición de conjuntos y variables es la siguiente:

### Conjuntos

$T$  Número de periodos en el horizonte de planificación indexado en  $t=1, \dots, T$

$K$  Número de centrales disponibles indexado en  $k=1, \dots, K$

$F$  Número de estados de cada central indexado en  $f=0, \dots, F-1$

### Variables

$x_{t,k,f}$  Denota si la central  $k$  se encuentra en el estado  $f$  en el periodo  $t$

$cr_{t,k,t,f}$  Coste asociado al cambio de estado (reconfiguración) de la central  $k$  en el periodo  $t$  debido al cambio de estado de  $f$  a  $f'$  en periodos consecutivos

## Planificación de la producción de electricidad en centrales térmicas (10)

La ecuación (5) del modelo original, ahora puede reemplazarse por la (5'). Nótese como el cambio del estado  $f$  al  $f'$  en cada central  $k$  para cada periodo  $t$  queda reflejado mediante la variable  $x$  en la ecuación (5a'). La ecuación (5b') es similar pero particularizada al primer periodo del horizonte de planificación. Nótese además, como dicho cambio de estado está multiplicado por el coste del mismo (coste de reconfiguración de la central  $k$  al pasar del estado  $f$  al  $f'$ ) para proporcionar directamente la variable  $cr$  a utilizar en la nueva función objetivo del problema, mostrada en la ecuación (1').

$$(5a') \quad cr_{t,k,f,f'} \geq CR_{k,f,f'} \cdot (x_{t-1,k,f} + x_{t,k,f'} - 1) \quad ; \quad \forall (t,k,f,f') : t > 1$$

$$(5b') \quad cr_{1,k,f,f'} \geq CR_{k,f,f'} \cdot (X0_{k,1,f} + x_{1,k,f'} - 1) \quad ; \quad \forall (k,f,f')$$

$$(1') \quad \min z = \sum_{t=1}^T \sum_{k=1}^K CPV_k \cdot p_{t,k} + \sum_{t=1}^T \sum_{k=1}^K \sum_{f=0}^{F-1} \sum_{f'=0}^{F-1} cr_{t,k,f,f'}$$

Nótese como se han agrupado los costes de procesamiento fijo ( $f=1$  ó  $0$  y  $f'=1$ ), de arranque ( $f=0$  y  $f'=1$ ), y de parada ( $f=1$  y  $f'=0$ ), en el coste de reconfiguración  $CR$ . Esta equivalencia se muestra en la ecuación siguiente, y debe tenerse en cuenta a la hora de introducir los datos del problema. Además, ahora el estado inicial de las centrales queda definido mediante el parámetro  $X0_{k,f}$  en lugar de  $X0_k$ .

$$CR_{k,0,0} = 0$$

$$CR_{k,0,1} = CS_k + CPF_k$$

$$CR_{k,1,0} = CE_k$$

$$CR_{k,1,1} = CPF_k$$

Finalmente, nótese que las ecuaciones (2) y (7), en las que aparece la variable  $x$ , deben ser ligeramente modificadas teniendo en cuenta la nueva definición de la misma.

$$(2') \quad PMIN_k \cdot x_{t,k,1} \leq p_{t,k} \leq PMAX_k \cdot x_{t,k,1} \quad ; \quad \forall (t,k)$$

$$(7') \quad \sum_{k=1}^K PMAX_k \cdot x_{t,k,1} \geq D_k + R_k \quad ; \quad \forall (t)$$



## Planificación de la producción de diferentes derivados en una refinería (1)

El proyecto propuesto consiste en el desarrollo de un modelo matemático con el que planificar a lo largo de un determinado horizonte de tiempo compuesto por varios periodos el tipo y cantidad de los diferentes derivados del petróleo que deben producirse en cada una de las plantas de procesado de una refinería para satisfacer la demanda de cada periodo. Las características que deben incluirse en el modelo son las siguientes:

- La planificación se llevará a cabo a lo largo de un horizonte de planificación discreto compuesto por  $T$  periodos.
- Se dispone de una refinería compuesta por  $K$  plantas de refinamiento.
- Cada una de dichas plantas es capaz de producir  $P$  productos diferentes.
- En cada periodo una planta puede estar parada o produciendo un único derivado.
- La cantidad a producir en cada periodo esta limitada tanto inferior como superiormente para cada planta y producto.
- Las cantidades de los diferentes derivados producidos en cada periodo pueden ser directamente destinadas a satisfacer la demanda del periodo correspondiente o almacenadas en tanques para satisfacer la demanda en periodos futuros.
- La capacidad de dichos tanques está limitada, existiendo además un límite inferior de reserva que asegure un stock mínimo para satisfacer algún imprevisto en la demanda.
- Existe la posibilidad de retrasar la entrega de una cierta cantidad de la demanda de cada periodo a periodos futuros. Dicho retraso puede ir acumulándose en los periodos sucesivos si en ellos la demanda sigue sin entregarse a tiempo, pero dicho retraso debe ser nulo para el último periodo del horizonte de planificación. Las entregas fuera de plazo tienen un coste asociado.
- El coste de producción es diferente para cada producto y para cada periodo de tiempo. Es decir, el procesado de algunos derivados es más costoso que el de otros y pueden existir ventanas de tiempo en las que el funcionamiento de las plantas sea más económico.
- El coste de inventario tan sólo depende del tipo de producto a almacenar.
- Existen costes asociados a la parada y puesta en marcha de cada planta. Dichos costes dependen de la planta y del producto que se esté produciendo antes de la parada o del producto que se vaya a producir tras el arranque.
- Cada planta puede cambiar el tipo de derivado a producir de un periodo a otro. El coste de tal reconfiguración depende de la planta y del par de productos involucrados en la misma.

## Planificación de la producción de diferentes derivados en una refinería (2)

- El coste asociado a la entrega de la demanda en un periodo posterior al especificado depende del tipo de producto demandado. Además, coste por unidad no entregada a tiempo no es constante, sino que depende de si el retraso en la entrega de la demanda ( $rd$ ) supera una determinada cantidad ( $RD$ ) o no según se muestra en la figura. Así, si en el periodo  $t$  existe una cantidad de producto  $p$ ,  $rd$ , que aun no ha sido entregada para satisfacer la demanda en los periodos anteriores, el coste asociado se puede escribir como:



$$Coste(rd) = \begin{cases} CIRD \cdot rd & \text{si } rd \leq RD \\ CIRD \cdot RD + CSRD \cdot (rd - RD) & \text{si } rd > RD \end{cases}$$

Así, debe buscarse alguna forma de modelar dicho coste mediante ecuaciones lineales.

- Finalmente, existe un beneficio por la venta del stock acumulado al final del horizonte de planificación, siendo el precio de venta diferente para cada producto

Así, el objetivo del problema es encontrar la secuencia y cantidad óptima de derivados a producir en cada una de las plantas de la refinería minimizando:

- OBJ1: El coste de producción.
- OBJ2: El coste de almacenamiento.
- OBJ3: El coste de arranques, paradas y reconfiguración de plantas.
- OBJ4: El coste asociado a las entregas fuera de plazo.

Y maximizando:

- OBJ5: Los ingresos por la venta de los productos en existencias al final del último periodo.

Como en los ejemplos anteriores de deben fijar claramente los conjuntos, parámetros y variables del problema para posteriormente establecer la función objetivo y restricciones del modelo matemático con el que obtener la solución al problema.

### Planificación de la producción de diferentes derivados en una refinería (3)

Se proporciona la definición completa de todos los conjuntos y parámetros del problema:

#### Conjuntos

- $T$  Número de periodos en el horizonte de planificación indexado en  $t=1,\dots,T$   
 $K$  Número de plantas de procesamiento disponibles indexado en  $k=1,\dots,K$   
 $P$  Número de productos a producir indexado en  $p=1,\dots,P$

#### Parámetros

- $X0_{k,p}$  Denota si el producto  $p$  se está produciendo en la planta  $k$  al inicio del horizonte de planificación. Si es cero para todo  $p$ , la planta  $k$  está parada.
- $A0_p$  Nivel de inventario de producto  $p$  al inicio del horizonte de planificación
- $AMIN_p$  Mínima capacidad de almacenamiento para el producto  $p$
- $AMAX_p$  Máxima capacidad de almacenamiento para el producto  $p$
- $CA_p$  Coste de almacenamiento del producto  $p$
- $CV_p$  Coste de venta del producto  $p$
- $D_{t,p}$  Demanda del producto  $p$  en el periodo  $t$
- $CP_{t,p}$  Coste de producción del producto  $p$  en el periodo  $t$
- $CMIN_{k,p}$  Mínima capacidad de producción del producto  $p$  en la planta  $k$
- $CMAx_{k,p}$  Máxima capacidad de producción del producto  $p$  en la planta  $k$
- $CS_{k,p}$  Coste de arranque de la planta  $k$  para producir el producto  $p$
- $CE_{k,p}$  Coste de parada de la planta  $k$  cuando estaba produciendo el producto  $p$
- $CR_{k,p,p'}$  Coste asociado a la reconfiguración de la planta  $k$  como consecuencia del procesado de productos diferentes  $p$  y  $p'$  en periodos consecutivos
- $CIRD_p$  Coste inferior asociado al retraso en la entrega de la demanda de producto  $p$
- $CSRd_p$  Coste superior asociado al retraso en la entrega de la demanda de producto  $p$
- $RD_p$  Máxima cantidad de producto  $p$  cuya entrega puede retrasarse a periodos posteriores antes de incrementar su coste  $CDs_p$

## Planificación de la producción de diferentes derivados en una refinería (4)

Todas las centrales están paradas al inicio del horizonte de planificación. Las siguientes tablas muestran los datos del problema:

Parámetro	P1 Gasolina	P2 Diesel	P3 LPG
AO	10	40	20
AMIN	10	10	10
AMAX	200	200	200
CA	2	3	1
CV	5	10	15

Parámetro	Demanda (D) / Coste de producción (CP)		
Periodo	P1 Gasolina	P2 Diesel	P3 LPG
1	90 / 10	80 / 20	50 / 15
2	70 / 10	40 / 20	40 / 15
3	30 / 15	80 / 30	40 / 20
4	30 / 15	30 / 30	70 / 20

Parámetro	CMIN / CMAX		
Planta	P1 Gasolina	P2 Diesel	P3 LPG
No 1	5 / 30	5 / 40	5 / 70
No 2	5 / 30	5 / 40	5 / 70
No 3	5 / 40	5 / 50	5 / 40
No 4	5 / 40	5 / 50	5 / 40

Parámetro	CS / CE		
Planta	P1 Gasolina	P2 Diesel	P3 LPG
No 1	4 / 0.4	3 / 0.3	5 / 0.5
No 2	4 / 0.4	3 / 0.3	5 / 0.5
No 3	4 / 0.4	3 / 0.3	5 / 0.5
No 4	4 / 0.4	3 / 0.3	5 / 0.5

Parámetro	CR (Planta No 1)		
	P1 Gasolina	P2 Diesel	P3 LPG
P1 Gasolina	×	50	30
P2 Diesel	50	×	20
P3 LPG	30	20	×

Parámetro	CR (Planta No 2)		
	P1 Gasolina	P2 Diesel	P3 LPG
P1 Gasolina	×	50	30
P2 Diesel	50	×	20
P3 LPG	30	20	×


Parámetro	CR (Planta No 3)		
	P1 Gasolina	P2 Diesel	P3 LPG
P1 Gasolina	×	50	30
P2 Diesel	50	×	20
P3 LPG	30	20	×

Parámetro	CR (Planta No 4)		
	P1 Gasolina	P2 Diesel	P3 LPG
P1 Gasolina	×	50	30
P2 Diesel	50	×	20
P3 LPG	30	20	×

Parámetro	P1 Gasolina	P2 Diesel	P3 LPG
CIRD	20	30	10
CSRD	50	80	40
RD	20	30	20

## Planificación de la producción de diferentes derivados en una refinería (5)

Ejecución en ILOG OPL Studio:



Final solution with objective 11516:

```
obj1 = 10750;
obj2 = 360;
obj3 = 106;
obj4 = 600;
obj5 = 300;
```

### Solución

Legenda:

- Producto 1: Gasolina
- Producto 2: Diesel
- Producto 3: LPG

Cantidad y tipo de producto a procesar en cada planta en cada periodo:

